///////////////////////////////////////////////////
// Election Administration Polarization
// Make RD Table with Lagged Outcomes
///////////////////////////////////////////////////


gl path "~/Library/CloudStorage/GoogleDrive-danmckinleythompson@gmail.com/My Drive/ElecAdminPolarization/How_Partisan_Is_Local_Election_Admin_Replication"


// Lagged Dem Presidential Vote Share

* Bring in the presidential election analysis data
use "$path/analysis_data/rd_election_analysis_data.dta", clear
keep if office=="pres"

rdbwselect lag_vs_dem rv, vce(cluster election_id) kernel(uniform)
gl bw = e(h_mserd)

* Spec 1, Residualized Outcome
reg lag_vs_dem treat rv treat_rv if abs(rv)<$bw, vce(clust election_id)
rdtab lag_vs_dem treat rv election_id, est_command("reg") new("B") tau(0.02) alpha(0.05) onesided

* Spec 2, Residualized Outcome
reg lag_vs_dem treat rv treat_rv if abs(rv)<$bw*2, vce(clust election_id)
rdtab lag_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 3, Residualized Outcome
reg lag_vs_dem treat rv treat_rv if abs(rv)<$bw*0.5, vce(clust election_id)
rdtab lag_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 4, Residualized Outcome
rdrobust lag_vs_dem rv, vce(cluster election_id) all
rdtab lag_vs_dem treat rv election_id, est_command("rdrobust") old("B") tau(0.02) alpha(0.05) onesided


// Lagged Turnout

* Bring in the turnout and registration analysis data
use "$path/analysis_data/rd_reg_turnout_analysis_data.dta", clear

rdbwselect lag_votes_highest_pc rv, vce(cluster election_id) kernel(uniform)
gl bw = e(h_mserd)

* Spec 1, Residualized Outcome
reg lag_votes_highest_pc treat rv treat_rv if abs(rv)<$bw, vce(clust election_id)
rdtab lag_votes_highest_pc treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 2, Residualized Outcome
reg lag_votes_highest_pc treat rv treat_rv if abs(rv)<$bw*2, vce(clust election_id)
rdtab lag_votes_highest_pc treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 3, Residualized Outcome
reg lag_votes_highest_pc treat rv treat_rv if abs(rv)<$bw*0.5, vce(clust election_id)
rdtab lag_votes_highest_pc treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 4, Residualized Outcome
rdrobust lag_votes_highest_pc rv, vce(cluster election_id) all
rdtab lag_votes_highest_pc treat rv election_id, est_command("rdrobust") old("B") tau(0.02) alpha(0.05) onesided


qui {
* Build the memo/paper table
cap log close
set linesize 255
log using "$path/output/rd_table_lag.tex", text replace
noi di "\begin{table}[ht]"
noi di "\centering"
noi di "\caption{\textbf{Regression Discontinuity Design Balances Pre-Treatment Democratic Presidential Vote Share and Turnout.}"
noi di "\label{tab:rd_table_lag}}"
noi di "\begin{tabular}{lcccc|cccc}"
noi di "\toprule \toprule"
noi di " & \multicolumn{4}{c}{Lagged Dem Pres Vote Share} & \multicolumn{4}{c}{Lagged Turnout}\\"
noi di " & (1) & (2) & (3) & (4) & (5) & (6) & (7) & (8) \\"
noi di "\midrule"
noi di "Dem Elec Official & " %4.3f B[1,1] " & " %4.3f B[1,2] " & " %4.3f B[1,3] " & " %4.3f B[1,4] " "
noi di 		" & " %4.3f B[1,5] " & " %4.3f B[1,6] " & " %4.3f B[1,7] " & " %4.3f B[1,8] "\\"
noi di " & (" %4.3f B[2,1] ") & (" %4.3f B[2,2] ") & (" %4.3f B[2,3] ") & (" %4.3f B[2,4] ") "
noi di 		" & (" %4.3f B[2,5] ") & (" %4.3f B[2,6] ") & (" %4.3f B[2,7] ") & (" %4.3f B[2,8] ") \\[2mm]"
noi di "\midrule"
noi di "N & " %4.0f B[3,1] " & " %4.0f B[3,2] " & " %4.0f B[3,3] " & " %4.0f B[3,4] " "
noi di 		" & " %4.0f B[3,5] " & " %4.0f B[3,6] " & " %4.0f B[3,7] " & " %4.0f B[3,8] " \\ "
noi di "Clusters & " %4.0f B[4,1] " & " %4.0f B[4,2] " & " %4.0f B[4,3] " & " %4.0f B[4,4] " "
noi di 		" & " %4.0f B[4,5] " & " %4.0f B[4,6] " & " %4.0f B[4,7] " & " %4.0f B[4,8] " \\ "
noi di "Bandwidth & " %4.2f B[5,1] " & " %4.2f B[5,2] " & " %4.2f B[5,3] " & " %4.2f B[5,4] " "
noi di 	" & " %4.2f B[5,5] " & " %4.2f B[5,6] " & " %4.2f B[5,7] " & " %4.2f B[5,8] " \\ "
noi di "BW Selection & CCT & CCT*2 & CCT/2 & CCT & CCT & CCT*2 & CCT/2 & CCT \\"
noi di "Kernel & Unif & Unif & Unif & Tri & Unif & Unif & Unif & Tri \\"
noi di "\bottomrule \bottomrule"
noi di "\multicolumn{9}{p{1\textwidth}}{\footnotesize Robust standard errors clustered by clerk election in parentheses. "
noi di "The bandwidth row reports the number of maximum clerk win margin allowed for inclusion in each specificaiton. "
noi di "CCT refers to \cite{calonico2014robust} bandwidth selection procedure. Unif means the specification uses a uniform kernel. "
noi di "Tri means the specification uses a triangular kernel.}"
noi di "\end{tabular}"
noi di "\end{table}"
log off
}

